Symbolic Integration

Victor Adamchik

Carnegie Mellon University

The Rioboo algorithm

Consider the following integral

$$\int \frac{x^2 + 2\,x + 4}{x^4 - 7\,x^2 + 2\,x + 17}\,dx$$

Since the denominator is square free

```
SquareFreeQ[17 + 2 x - 7 x^2 + x^4, x]
```

```
True
```

we proceed with Trager's algorithm

```
p = 4 + 2 x + x^2;
q = 17 + 2 x - 7 x^2 + x^4;
Resultant[q, p - z D[q, x], x] // Factor
```

```
2169 (1 + 4 z^2)^2
```

```
roots = Union[z /. Solve[% == 0, z]];
args = PolynomialGCD[q, p - # D[q, x]] & /@ roots;
res = Inner[#1 Log[#2] &, roots, args, Plus]
```

$$-\frac{1}{2}\,i\,\mathrm{Log}\left[(-4 - i) - i\,x + x^2\right] + \frac{1}{2}\,i\,\mathrm{Log}\left[(-4 + i) + i\,x + x^2\right]$$

However, the integral can be computed without the complex numbers extension

```
Integrate[(4 + 2 x + x^2)/(17 + 2 x - 7 x^2 + x^4), x]
```

```
1/2 ArcTan[(-1 - x)/(-4 + x^2)] - 1/2 ArcTan[(1 + x)/(-4 + x^2)]
```

■ Idea 1 - Log to Arctan

Given a field K of characteristic 0 such that $\sqrt{-1} \notin K$, and $A, B \in K[x]$ with $B \neq 0$, return a sum of arctangents defined by

$$\frac{d}{dx}\arctan\left(\frac{A}{B}\right) = \frac{d}{dx}\frac{i}{2}\log\left(\frac{A+iB}{A-iB}\right)$$

We will be using $\overset{D}{=}$ to denote that computations are correct under differentiation (or up to a constant). Indeed,

$$\arctan\left(\frac{A}{B}\right) \overset{D}{=} \frac{i}{2}\log\left(\frac{A+iB}{A-iB}\right)$$

We use differentiation to eliminate the constant of integration

```
A = 1.234; B = 0.432; ArcTan[A/B] - I/2 Log[(A + I B)/(A - I B)]
```

```
1.5708 + 0. i
```

```
A = -1.234; B = 0.432; ArcTan[A/B] - I/2 Log[(A + I B)/(A - I B)]
```

```
-1.5708 + 0. i
```

The identity readily follows from the definition of the arctangent function

$$\arctan(x) = \frac{i}{2}\log(1 - i\,x) - \frac{i}{2}\log(1 + i\,x) \overset{D}{=} \frac{i}{2}\log\left(\frac{1 - i\,x}{1 + i\,x}\right)$$

Indeed,

$$\arctan\left(\frac{A}{B}\right) \overset{D}{=} \frac{i}{2}\log\left(\frac{1 - i\,\frac{A}{B}}{1 + i\,\frac{A}{B}}\right) = \frac{i}{2}\log\left(\frac{B - iA}{B + iA}\right) = \frac{i}{2}\log\left(\frac{\frac{B}{i} - A}{\frac{B}{i} + A}\right) = \frac{i}{2}\log\left(\frac{-iB - A}{-iB + A}\right) \overset{D}{=} \frac{i}{2}\log\left(\frac{A + iB}{A - iB}\right)$$

Using this idea we can rewrite the result of integration as follows

$$\frac{i}{2}\log\left(x^2+i\,x+(-4+i)\right)-\frac{i}{2}\log\left(x^2-i\,x+(-4-i)\right)\overset{D}{=}\frac{i}{2}\log\left(\frac{x^2+i\,(x+1)-4}{x^2-i\,(x+1)-4}\right)\overset{D}{=}\arctan\left(\frac{x^2-4}{x+1}\right)$$

Therefore,

$$\int\frac{x^2+2\,x+4}{x^4-7\,x^2+2\,x+17}\,dx=\arctan\left(\frac{x^2-4}{x+1}\right)+\,C$$

```
D[ArcTan[(x^2 - 4)/(x + 1)], x] // Simplify
```

```
(4 + 2 x + x^2)/(17 + 2 x - 7 x^2 + x^4)
```

which matches *Mathematica*'s result up to the constant of integration

```
Integrate[(4 + 2 x + x^2)/(17 + 2 x - 7 x^2 + x^4), x] // Simplify
```

```
ArcTan[(1 + x)/(4 - x^2)]
```

Note,

```
D[ArcTan[x] + ArcTan[1/x], x] // Together
```

```
0
```

It follows, that both results are correct under differentiation

$$\int\frac{x^2+2\,x+4}{x^4-7\,x^2+2\,x+17}\,dx=\arctan\left(\frac{x^2-4}{x+1}\right)+\,C_1=\arctan\left(\frac{x+1}{4-x^2}\right)+\,C_2$$

■ Definite Integration

The proposed simplification causes computational problems for a definite integration, since an arctangent of a rational function contains hidden singularities

```
Plot[ArcTan[(x + 1)/(4 - x^2)], {x, 1, 3}]
```



We all know that proper definite integrals are evaluated by means of the Newton-Leibniz theorem

$$\int_a^b f(x)\,dx = F(b) - F(a)$$

where $F(x)$ is an antiderivative. In case of the integral discussed above, we get

$$\int_0^4 \frac{x^2+2x+4}{x^4-7x^2+2x+17}\,dx = \arctan\left(\frac{x+1}{4-x^2}\right)\Big|_{x=4} - \arctan\left(\frac{x+1}{4-x^2}\right)\Big|_{x=0}$$

And this is simply <u>wrong</u>

```
res = ArcTan[(x + 1)/(4 - x^2)];
(res /. x → 4) - (res /. x → 0) // N
```

```
-0.63977
```

```
NIntegrate[(x^2 + 2 x + 4)/(x^4 - 7 x^2 + 2 x + 17), {x, 0, 4}]
```

```
2.50182
```

The problem is that the path of integration (0, 4) crosses singularities of the antiderivative. The following picture demonstrates this

```
ContourPlot[Evaluate[Re[ArcTan[(x + 1)/(4 - x^2)] /. x -> x + I y]],
 {x, -3, 4}, {y, -3, 4}, ContourShading -> False,
 Contours -> 20, PlotPoints -> 40,
 Epilog -> {Hue[0], Thickness[0.005], Line[{{0, 0}, {4, 0}}],
   RGBColor[0, 0, 0], Thickness[0.003],
   Line[{{1.95, -0.756}, {1.95, 0.756}}],
   Line[{{-1.95, -0.246}, {-1.95, 0.246}}]
  }]
```



Observe that *Mathematica* computes the definite integral correctly

```
Integrate[(x^2 + 2 x + 4)/(x^4 - 7 x^2 + 2 x + 17), {x, 0, 4}] // N
```

```
2.50182
```

How does *Mathematica* do it? By computing limits at singularities

```
Limit[res, x → 4, Direction ->  1] -
 Limit[res, x → 2, Direction -> -1] +
 Limit[res, x → 2, Direction ->  1] -
 Limit[res, x → 0, Direction -> -1] // N
```

```
2.50182
```

There is a better idea! We can convert logs into arctangents of polynomials.

■ Idea 2

The "identity" (under differentiation) we used in the previous section

$$\arctan\left(\frac{A}{B}\right) \stackrel{D}{=} \frac{i}{2}\log\left(\frac{A+iB}{A-iB}\right)$$

can be further expanded

$$\arctan\left(\frac{A}{B}\right) \stackrel{D}{=} \frac{i}{2}\log\left(\frac{A+iB}{A-iB}\right) \stackrel{D}{=} \arctan\left(\frac{A\,d+B\,c}{g}\right) + \frac{i}{2}\log\left(\frac{d+i\,c}{d-i\,c}\right)$$

where

$$g = B\,d - A\,c,$$

$$\deg(d) < \deg(A), \;\; \deg(c) < \deg(A)$$

Proof. We will simplify $\log\left(\frac{d+i\,c}{d-i\,c}\right)$ by using

$$\arctan\left(\frac{A}{B}\right) \stackrel{D}{=} \frac{i}{2}\log\left(\frac{A+iB}{A-iB}\right)$$

We get

$$\frac{i}{2}\log\left(\frac{d+i\,c}{d-i\,c}\right) \stackrel{D}{=} \arctan\left(\frac{d}{c}\right)$$

Thus,

$$\arctan\left(\frac{A\,d+B\,c}{B\,d-A\,c}\right) + \frac{i}{2}\log\left(\frac{d+i\,c}{d-i\,c}\right) \stackrel{D}{=} \arctan\left(\frac{A\,d+B\,c}{B\,d-A\,c}\right) + \arctan\left(\frac{d}{c}\right)$$

Next, we apply the summation formula for arctangents

$$\arctan(x) + \arctan(y) = \arctan\left(\frac{x+y}{1-x\,y}\right)$$

```
Clear[A, B, c, d];
x = (A d + B c)/(B d - A c); y = d/c; Simplify[(x + y)/(1 - x y)]
```

```
-B/A
```

We obtain,

$$\arctan\left(\frac{A\,d+B\,c}{g}\right)+\arctan\left(\frac{d}{c}\right)\overset{D}{=}\arctan\left(-\frac{B}{A}\right)\overset{D}{=}\arctan\left(\frac{A}{B}\right)$$

whih concludes the proof. QED

■ Example

Consider the result of Trager's algorihm

$$\int\frac{x^2+2\,x+4}{x^4-7\,x^2+2\,x+17}\,dx=\frac{i}{2}\log\left(\frac{x^2-4+i\,(x+1)}{x^2-4-i\,(x+1)}\right)+\;C$$

We will convert the complex log function into a sum of arctangents by recursively applying this

$$\frac{i}{2}\;\log\left(\frac{A+i\,B}{A-i\,B}\right)\overset{D}{=}\arctan\left(\frac{A\,d+B\,c}{g}\right)+\frac{i}{2}\;\log\left(\frac{d+i\,c}{d-i\,c}\right)$$

First we find g by running the EEA

```
Clear[x];
A = x^2 - 4;
B = x + 1;
{g, {d, c}} = PolynomialExtendedGCD[B, -A, x]
```

```
{1, {1/3 (-1 + x), 1/3}}
```

and then compute

```
ArcTan[(A d + B c)/g // Simplify] + I/2 Log[(d + I c)/(d - I c)]
```

$$\mathrm{ArcTan}\left[\frac{1}{3}\left(5-3x-x^2+x^3\right)\right]+\frac{1}{2}\,i\,\mathrm{Log}\left[\frac{\frac{i}{3}+\frac{1}{3}(-1+x)}{-\frac{i}{3}+\frac{1}{3}(-1+x)}\right]$$

to get the following

$$\frac{i}{2}\log\left(\frac{x^2-4+i(x+1)}{x^2-4-i(x+1)}\right)\stackrel{D}{=}\arctan\left(\frac{1}{3}\left(x^3-x^2-3x+5\right)\right)+\frac{i}{2}\log\left(\frac{\frac{x-1}{3}+\frac{i}{3}}{\frac{x-1}{3}-\frac{i}{3}}\right)$$

Apply the same procedure to the remaining log

```
A = (x - 1)/3; B = 1/3;
{g, {d, c}} = PolynomialExtendedGCD[B, -A, x];
ArcTan[(A d + B c)/g // Simplify] + I/2 Log[(d + I c)/(d - I c)]
```

```
-ArcTan[1 - x]
```

we get

$$\frac{i}{2}\log\left(\frac{\frac{x-1}{3}+\frac{i}{3}}{\frac{x-1}{3}-\frac{i}{3}}\right)\stackrel{D}{=}-\arctan(1-x)$$

Combining steps together, yields

$$\frac{i}{2}\log\left(\frac{x^2-4+i(x+1)}{x^2-4-i(x+1)}\right)\stackrel{D}{=}\arctan\left(\frac{1}{3}\left(x^3-x^2-3x+5\right)\right)-\arctan(1-x)$$

and

$$\int\frac{x^2+2x+4}{x^4-7x^2+2x+17}\,dx=\arctan\left(\frac{1}{3}\left(x^3-x^2-3x+5\right)\right)-\arctan(1-x)\;+C$$